What surrounds my Minesweeper tile? Minesweeper is a puzzle game where mines are hidden around a board of nondescript tiles with the goal of identifying the location of all mines. Clicking on a mine loses the game, but clicking on any other tile will reveal a number from 0-8 which signifies how many mines directly surround it.

Given a number, you must display a random* possible combination of empty tiles and mines surrounding it. This should be in the form of a 3x3 array. The center tile should be the number of mines taken as input.

*Must have a non-zero chance for all combinations to occur.



Examples

_ = blank square
X = mine

0

___
_0_
___

1

_X_
_1_
___

1

___
_1_
X__

___
_1_
__X

4

_X_
X4X
_X_

4

X_X
_4_
X_X

4

___
X4X
X_X

8

XXX
X8X
XXX




Input


The number of mines surrounding the center tile (0-8)




Output


Any reasonable form of output that displays the 3x3 array of tiles




Other rules


Each combination does not have to have an equal chance of happening. There just must be a non-zero chance of each combination to occur when executing your program.
Any 2 characters can be chosen for the mine and empty tile.
This is code golf, the program with fewest bytes wins.
APL (Dyalog Unicode), 28 15 bytes



-13 bytes thanks to ngn!

{3 3⍴5⌽⍵,⍵≥8?8}


Explanation:

{...} A direct function (D-Fn), ⍵ is its right argument.

8?8 deal 8 random numbers from the list 1..8:

      8?8                         
7 2 1 8 4 6 5 3


⍵≥ is the argument greater or equal to each of them?:

      5 ≥ 7 2 1 8 4 6 5 3   
0 1 1 0 1 0 1 1


⍵, prepend the argument to the boolean list:

      5 , 0 1 1 0 1 0 1 1
5 0 1 1 0 1 0 1 1


5⌽ rotate the list 5 positions to the left, so that the argument is in the center:

      5 ⌽ 5 0 1 1 0 1 0 1 1
1 0 1 1 5 0 1 1 0


3 3⍴ reshape the list to a 3x3 matrix:

      3 3 ⍴ 1 0 1 1 5 0 1 1 0
1 0 1
1 5 0
1 1 0


Try it online!

J, 15 bytes

Also many bytes thanks to ngn!

3 3$5|.],]&gt;8?8:


Try it online!